<commit_before>" MIT License. Copyright (c) 2013-2014 <NAME>.
" vim: et ts=2 sts=2 sw=2

let s:prototype = {}

function! s:prototype.split(...)
  let self._side = 0
  let self._line .= '%#'.self._curgroup.'#'.(a:0 ? a:1 : '%=')
endfunction

function! s:prototype.add_section_spaced(group, contents)
  call self.add_section(a:group, (g:airline_symbols.space).a:contents.(g:airline_symbols.space))
endfunction

function! s:prototype.add_section(group, contents)
  if self._curgroup != ''
    if self._curgroup == a:group
      let self._line .= self._side ? self._context.left_alt_sep : self._context.right_alt_sep
    else
      call airline#highlighter#add_separator(self._curgroup, a:group, self._side)
      let self._line .= '%#'.self._curgroup.'_to_'.a:group.'#'
      let self._line .= self._side ? self._context.left_sep : self._context.right_sep
    endif
  endif

  if self._curgroup != a:group
    let self._line .= '%#'.a:group.'#'
  endif

  if self._context.active
    let contents = []
    let content_parts = split(a:contents, '__accent')
    for cpart in content_parts
      let accent = matchstr(cpart, '_\zs[^#]*\ze')
      call add(contents, cpart)
    endfor
    let line = join(contents, a:group)
    let line = substitute(line, '__restore__', a:group, 'g')
  else
    let line = substitute(a:contents, '%#__accent[^#]*#', '', 'g')
    let line = substitute(line, '%#__restore__#', '', 'g')
  endif

  let self._line .= line
  let self._curgroup = a:group
endfunction

function! s:prototype.add_raw(text)
  let self._line .= a:text
endfunction

function! s:prototype.build()
  if !self._context.active
    let self._line = substitute(self._line, '%#.\{-}\ze#', '\0_inactive', 'g')
  endif
  return self._line
endfunction

function! airline#builder#new(context)
  let builder = copy(s:prototype)
  let builder._context = a:context
  let builder._side = 1
  let builder._curgroup = ''
  let builder._line = ''

  call extend(builder._context, {
        \ 'left_sep': g:airline_left_sep,
        \ 'left_alt_sep': g:airline_left_alt_sep,
        \ 'right_sep': g:airline_right_sep,
        \ 'right_alt_sep': g:airline_right_alt_sep,
        \ }, 'keep')
  return builder
endfunction
<commit_msg>refactor builder and reduce state across multiple calls<commit_after>" MIT License. Copyright (c) 2013-2014 <NAME>.
" vim: et ts=2 sts=2 sw=2

let s:prototype = {}

function! s:prototype.split(...)
  call add(self._sections, ['|', a:0 ? a:1 : '%='])
endfunction

function! s:prototype.add_section_spaced(group, contents)
  call self.add_section(a:group, (g:airline_symbols.space).a:contents.(g:airline_symbols.space))
endfunction

function! s:prototype.add_section(group, contents)
  call add(self._sections, [a:group, a:contents])
endfunction

function! s:prototype.add_raw(text)
  call add(self._sections, ['', a:text])
endfunction

function! s:prototype.build()
  let side = 1
  let prev_group = ''
  let line = ''

  for section in self._sections
    let group = section[0]
    let contents = section[1]

    if group == '|'
      let side = 0
      let line .= contents
      continue
    endif

    if prev_group != ''
      if prev_group == group
        let line .= side ? self._context.left_alt_sep : self._context.right_alt_sep
      elseif group != ''
        call airline#highlighter#add_separator(prev_group, group, side)
        let line .= '%#'.prev_group.'_to_'.group.'#'
        let line .= side ? self._context.left_sep : self._context.right_sep
      endif
    endif

    if group != prev_group
      let line .= '%#'.group.'#'
    endif
    let line .= s:get_accented_line(self, group, contents)

    if group != ''
      let prev_group = group
    endif
  endfor

  if !self._context.active
    let line = substitute(line, '%#.\{-}\ze#', '\0_inactive', 'g')
  endif
  return line
endfunction

function! s:get_accented_line(self, group, contents)
  if a:self._context.active
    let contents = []
    let content_parts = split(a:contents, '__accent')
    for cpart in content_parts
      let accent = matchstr(cpart, '_\zs[^#]*\ze')
      call add(contents, cpart)
    endfor
    let line = join(contents, a:group)
    let line = substitute(line, '__restore__', a:group, 'g')
  else
    let line = substitute(a:contents, '%#__accent[^#]*#', '', 'g')
    let line = substitute(line, '%#__restore__#', '', 'g')
  endif
  return line
endfunction

function! airline#builder#new(context)
  let builder = copy(s:prototype)
  let builder._context = a:context
  let builder._sections = []

  call extend(builder._context, {
        \ 'left_sep': g:airline_left_sep,
        \ 'left_alt_sep': g:airline_left_alt_sep,
        \ 'right_sep': g:airline_right_sep,
        \ 'right_alt_sep': g:airline_right_alt_sep,
        \ }, 'keep')
  return builder
endfunction

{'fraction_non_alphanumeric': 0.14581157301023606, 'fraction_numerical': 0.007102569458951326, 'mean_word_length': 3.453953488372093, 'pattern_counts': {'":': 0, '<': 5, '<?xml version=': 0, '>': 5, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5367645', 'n_tokens_mistral': 1678, 'n_tokens_neox': 1625, 'n_words': 372}